#============================================
# 「技能基类」 -  2020-9-12 22:23
# @author: Apprentice Zhang
#============================================

extends Node2D
class_name SkillBase


#>>> 枚举 <<<
# 技能状态类型
enum SKILL_STATE{
	IDLE			# 空闲，可施放的
	CASTED			# 已经施放
	CASTING			# 正在施放
	FINISHED		# 技能施放完成
}


#>>> 面板属性 <<<
export (NodePath) var host_node = @'..'
export (String) var skill_name					# 技能名称
export (Texture) var skill_icon					# 技能图标
export (String) var input_key					# 施放键（获取「键位映射」里的）
export (float, 0, 10000) var cooldown = 1.0		# 冷却时间
export (float, 0, 10000) var duration = 0.5		# 技能持续时间


#>>> 私有变量 <<<
var _skill_state: int = SKILL_STATE.IDLE		# 当前技能状态
var _can_cast: bool = true						# 能否施放技能
var _duration_countdown: float = -1				# 施放持续技能倒计时


#>>> 节点 <<<
var _cooling_timer := Timer.new()				# 技能冷却计时器



#============================================
#		内置的方法
#============================================
func _ready() -> void:
	set_physics_process(false)
	
	if input_key == '':
		print("--> SkillBase < 提示 > 未设置按键")
		set_process(false)
	
	# 添加「冷却时间」计时器
	if not _cooling_timer.is_inside_tree():
		_cooling_timer.connect("timeout", self, 'cooling_complete')
		_cooling_timer.one_shot = true
		_cooling_timer.autostart = false
		add_child(_cooling_timer)


func _unhandled_input(event: InputEvent) -> void:
	# 获取用户按键
	if event.is_action_pressed(input_key):
		if is_can_cast():
			cast_skill()


func _physics_process(delta: float) -> void:
	# 「持续技能倒计时」到达时间，则关闭进程
	if _duration_countdown <= 0:
		cast_over()
		set_physics_process(false)
		return
	_duration_countdown -= delta	# 「持续技能时间倒计时」
	duration_process(delta)			# 调用持续技能方法



#============================================
#		可重写方法
#============================================
func cast_skill():
	"""施放技能"""
#	print("--> 开始施放技能：", skill_name)
	_start_duration_countdown()
	_start_cooldown_countdown()
	_can_cast = false
	switch_skill_state(SKILL_STATE.CASTED)


func duration_process(delta: float):
	"""持续时间进程"""
#	print('---> ', skill_name)
	pass


func cast_over():
	"""技能施放结束"""
#	print("--> 技能施放结束：", skill_name)
	if _skill_state != SKILL_STATE.IDLE:
		switch_skill_state(SKILL_STATE.FINISHED)


func is_can_cast() -> bool:
	"""能否施放技能
	< 后续重写时，可再添加魔法值等属性是否足够的判断，来确定可否施放技能 >
	"""
	return _can_cast && _skill_state == SKILL_STATE.IDLE


func cooling_complete() -> void:
	"""技能冷却完成"""
	switch_skill_state(SKILL_STATE.IDLE)
	_can_cast = true


func switch_skill_state(state):
	"""切换技能状态"""
	_skill_state = state



#============================================
#		最终的方法（不能重写的方法）
#============================================
func _start_duration_countdown() -> void:
	"""开启持续技能倒计时"""
	if duration <= 0:
		cast_over()
		return
	switch_skill_state(SKILL_STATE.CASTING)
	_duration_countdown = duration
	set_physics_process(true)


func _start_cooldown_countdown() -> void:
	"""开始冷却时间倒计时"""
	if cooldown > 0 :	_cooling_timer.start(cooldown)
	else:				_cooling_timer.start(0.001)


func get_timer_once(
	wait_time: float,
	target: Object = null,
	method: String = ""
) -> Timer:
	"""返回一个一次性的计时器
	< 计时器结束后调用指定的方法，用于那些需要间隔时间施放的「立即施放」技能 >
	"""
	var _timer = Timer.new()
	if wait_time <= 0:	_timer.wait_time = 0.001
	else:				_timer.wait_time = wait_time
	_timer.connect("timeout", self, '__on_Timer_timeout', [_timer])
	_timer.autostart = true
	_timer.one_shot = true
	if target != null: 
		_timer.connect("timeout", target, method)
	add_child(_timer)
	return _timer


func __on_Timer_timeout(timer: Timer) -> void:
	timer.queue_free()


func _get_host_node() -> Node2D:
	"""获取节点的宿主"""
	return get_node(host_node) as Node2D
